# Maintainer: Carsten Haitzler <raster@archlinux.org>

pkgname=ephoto
pkgver=1.6.0
pkgrel=2
pkgdesc="EFL photo gallery viewer"
arch=('loong64' 'x86_64')
url="https://www.enlightenment.org/"
license=('BSD')
depends=('efl' 'libexif')
makedepends=('meson' 'ninja')
source=("https://download.enlightenment.org/rel/apps/${pkgname}/${pkgname}-${pkgver}.tar.xz")
sha256sums=('36bc73f3ce1bf9c606630d0031d1629a115911e78b796fd1f90322b5886670d3')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  export CFLAGS="${CFLAGS} -fvisibility=hidden"
  meson build --prefix=/usr
  ninja -C build
}

package(){
  cd "${srcdir}/${pkgname}-${pkgver}"

  DESTDIR="${pkgdir}" ninja -C build install

  # install license files
  install -Dm644 -t "$pkgdir/usr/share/licenses/$pkgname/" AUTHORS
}
